


Arquitectura de Computadores

Clase 5 - Saltos y Subrutinas

Profesor: Germán Leandro Contreras Sagredo

Bibliografía

- Apuntes históricos. Hans Löbel, Alejandro Echeverría
 - 06 Saltos y Subrutinas
- D. Patterson, Computer Organization and Design RISC-V Edition: The Hardware Software Interface. Morgan Kaufmann, 2020.
 - Capítulos 2.7-2.8. Página 92, 131 en PDF.
 - Capítulos 4.1-4.2-4.3-4.4. Página 236,
 320 en PDF.*



^{*} Estos se basan en la implementación de una arquitectura que implementa RISC-V. Tomar solo como referencia.


Objetivos de la clase

- Entender el impacto a nivel de código de habilitar saltos y subrutinas en el computador básico.
- Conocer los componentes necesarios para habilitar saltos y subrutinas en el computador básico.
- Realizar ejercicios que consoliden los conocimientos anteriores.

Hasta ahora...

En la figura se ve que tenemos lo necesario para:

- Operar con constantes numéricas (literales).
- Almacenar variables.
- Operar con direccionamiento directo e indirecto.




Hasta ahora...

No obstante lo anterior:


- ¿Podemos ejecutar iteraciones? (for, while)
- ¿Podemos definir funciones y ejecutarlas?

Al otorgar esta capacidad a nuestra máquina, esta será realmente programable.




Entenderemos por "salto" la capacidad de ejecutar, a partir de una instrucción, una línea de código de nuestro programa que **no necesariamente** sea la siguiente de la secuencia.

El componente que debemos modificar para este fin es el *Program Counter*.



En primer lugar, se añade la señal L_{PC} para que el contador PC pueda actualizar su valor según el dato de entrada. En este caso, corresponderá al **literal** de la instrucción.

El literal será la dirección de memoria de la instrucción que deseamos ejecutar en el siguiente ciclo.



 * En este caso, la señal $L_{\rm pc}$ tiene prioridad por sobre la señal Up del contador PC, lo que permite que, aunque ambas se activen, el resultado sea la carga del bus de entrada y no el incremento del valor actual.

El hardware ahora permite definir la instrucción JMP label, que realiza un salto incondicional a la primera instrucción de código asociada a label. En el ejemplo, se realiza un salto incondicional hacia test, lo que deriva en que el registro A almacene el valor 8 y no 12.

MOV A,2

JMP test

MOV A,4

test:

ADD A,2

SHL A,A

El **Assembler** se encarga de hacer la traducción correcta del *label* (en este caso, test) al literal correspondiente a la dirección de la primera instrucción a la que se asocia (en este caso, ADD A, 2).

Consideremos ahora el siguiente código (pseudocódigo en Python). En este caso, se lleva a cabo la iteración dentro del while si, y solo si la variable c cumple con ser mayor a cero. Es decir, si se cumple la condición, realizamos un salto a las instrucciones que se encuentran dentro del while.


```
a = 2
b = 3
c = 2
while (c > 0):
    a += b
    c -= 1
```

Estos son saltos condicionales y veremos cómo incluirlos en nuestros programas a través de modificaciones en el *hardware*.

Para realizar saltos condicionales, necesitamos contar con la capacidad de revisar condiciones.


Estas condiciones pueden provenir de una operación. Por ejemplo: a > b == a - b > 0

La **ALU** nos puede ayudar a evaluar condiciones **a partir del resultado de su operación**.



De la ALU ahora obtenemos *flags* (señales de 1 bit) que nos indican el cumplimiento o no de una condición. Particularmente:


- **Z: Z**ero. Indica si el resultado de la operación fue cero o no.
- N: Negative. Indica si el resultado de la operación fue negativo o no.



Asumimos que la ALU, en su circuito interno, se encarga de la obtención de estas señales. Por ejemplo, la *flag N* se puede deducir del bit más significativo del número resultante, mientras que la *flag Z* se puede obtener a partir de un circuito que verifique todos los bits sean iguales a cero (compuertas OR en cadena o pirámide, por ejemplo).

Las *flags* se almacenan en un nuevo registro *Status* conectado al *clock* del sistema para poder evaluar la condición sin que cambie durante un ciclo de ejecución. Siempre se actualiza, no depende de una señal *Load*.


La Control Unit evalúa la condición y realiza el salto o no a través del valor de la señal L_{PC}.



Si no se cumple la condición, la unidad de control propaga la señal $L_{PC}=0$ para asegurar que se ejecute la siguiente instrucción y no la indicada por el literal. Si se cumple, se propaga la señal $L_{PC}=1$ y se habilita el salto al sobreescribir la dirección de la instrucción a ejecutar.


Por ahora, estamos habilitando saltos en caso de que el resultado de la operación de la ALU sea negativo o cero.

¿Qué otras condiciones podríamos incluir?



- C: Carry. Indica si hubo un bit de carry al finalizar la operación.
- O: Overflow. Indica si hubo o no overflow al finalizar la operación.

Ahondaremos un poco más en estas nuevas *flags*.



Computador básico - Flags C y 0

Flag C: Es 1 si hubo bit de carry en una operación ADD/SUB, pero también se utiliza para representar el bit "descartado" en operaciones tipo shift.

Ejemplo: SHR(00101011) \rightarrow C = 1

Flag 0: Es 1 si hubo overflow en operaciones ADD/SUB. Algunas implementaciones también lo consideran en operaciones tipo shift, pero no todas. Existen cuatro casos a considerar:

- $A \ge 0$; $B \ge 0 \rightarrow 0 = 1$ si A + B < 0
- A < 0; $B < 0 \rightarrow 0 = 1$ si $A + B \ge 0$
- $A \ge 0$: $B < 0 \rightarrow 0 = 1$ si A B < 0
- A < 0; $B \ge 0 \rightarrow 0 = 1$ si $A B \ge 0$

Para computar las *flags* antes de un salto condicional, definimos las instrucciones CMP que tienen las siguientes señales de control:

- $L_A = L_B = 0$ (no cargamos datos en los registros).
- S_{Op} = SUB (operación resta).

De esta forma, se actualiza el registro *Status* a partir de la operación *A - B* sin actualizar el valor de estos. La instrucción siguiente a CMP debe ser de salto condicional para poder evaluar las *flags* de la operación y realizar o no la carga sobre el registro PC.

A continuación, las instrucciones de salto disponibles.

- JEQ: Jump equal. Se realiza el salto si Z = 1 (A = B).
- JNE: Jump not equal. Se realiza el salto si Z = 0 ($A \neq B$).
- **JGT:** Jump greater than. Se realiza el salto si N = 0 y Z = 0 (A > B).
- JLT: Jump less than. Se realiza el salto si N = 1 (A < B).

A continuación, las instrucciones de salto disponibles.

- **JGE:** Jump greater or equal than. Se realiza el salto si N = 0 $(A \ge B)$.
- JLE: Jump less or equal than. Se realiza el salto si N = 1 o Z = 1 $(A \le B)$.
- JCR: Jump carry. Se realiza el salto si C = 1 (hubo carry).
- **JOV:** Jump overflow. Se realiza el salto si O = 1 (hubo overflow).

Computador básico - Assembly de saltos

Instrucción	Operandos	Operación	Condición	Ejemplo
	A,B	A - B	-	
	A,(B)	A - Mem[B]	-	
CPM	A,Lit	A - Lit	-	CMP A,0
	A,(Dir)	A - Mem[Dir]	-	CMP A,(label)
ЈМР	Dir	PC = Dir	-	JMP label
JEQ	Dir	PC = Dir	Z = 1	JEQ label
JNE	Dir	PC = Dir	Z = 0	JNE label
JGT	Dir	PC = Dir	Z = 0 y N = 0	JGT label
JLT	Dir	PC = Dir	N = 1	JLT label
JGE	Dir	PC = Dir	N = 0	JGE label
JLE	Dir	PC = Dir	Z = 1 o N = 1	JLE label
JCR	Dir	PC = Dir	C = 1	JCR label
JOV	Dir	PC = Dir	0 = 0	JOV label

Ahora que tenemos a nuestra disposición las instrucciones de comparación y saltos, revisemos cómo queda el código de la iteración while en Assembly.

```
a = 2
b = 3
c = 2
while (c > 0):
    a += b
    c -= 1
```

Programa en pseudocódigo (Python).

```
DATA:
  a 2
  b 3
  c 2
CODE:
  MOV A, (c)
  MOV B, (b)
  while:
    CMP A,0
    JLE end
    SUB A,1
    MOV(c),A
    MOV A_{3}(a)
    ADD A,B
    MOV(a),A
    MOV A, (c)
    JMP while
  end:
    MOV A_{3}(a)
```

Programa en Assembly.

```
DATA:
  fibOne 1
                  ;Fibonacci N
 fibTwo 0
                   :Fibonacci N-1
CODE:
  loop:
   MOV A,(fibOne) ;A = Mem[fibOne]
   MOV B,(fibTwo) ;B = Mem[fibTwo]
   MOV (fibTwo), A ; Mem[fibTwo] = A = Fibonacci N
   ADD A,B ;A += B = Fibonacci N+1
   JOV display ;0 = 1 \rightarrow PC = display
   MOV (fibOne),A ;Mem[fibOne] = A = Fibonacci N+1
   JMP loop
  display:
   MOV A, (fibTwo) ; A = Último elemento válido
   MOV B,0
  end:
    JMP end
```

Ahora, podemos crear un programa más sofisticado para la obtención de los números de Fibonacci. Se registran los últimos números de la secuencia en las variables fib0ne y fibTwo según el poder de representación del computador, lo que se determina por overflow.

```
DATA:
 arr
                   ; arr = [1, 3, 5]
                   ;len = len(arr)
  len
 index 0
                   ;index = arr[index]
CODE:
 MOV B, arr
                  ;B = arr = dirección memoria arr
 loop:
   MOV A, (B) ;A = Mem[B] = Mem[arr]
   SHL A, A ; A = SHL A = A * 2
   MOV (B),A ;Mem[B] = A = Mem[arr] * 2
MOV A,(index) ;A = Mem[index]
   ADD A,1
            ;A += 1
   CMP A, (len) ;A - Mem[len]
   JGE end ;N = 0 \rightarrow PC = end
   MOV (index),A ;Mem[index] = A = index + 1
              ;B += 1 = dirección elemento arr + 1
   INC B
   JMP loop
 end:
    JMP end
```

También podemos recorrer un arreglo hasta pasar por todos sus elementos. En este caso, duplicamos todos los elementos y terminamos la ejecución cuando el índice es mayor o igual al largo del arreglo (i.e. while index < len).

```
DATA:
                   ; arr = [1, 3, 5]
 arr
                   :len = len(arr)
 len
 index 0
                   ;index = arr[index]
CODE:
 MOV A, (index)
                   ;A = Mem[index] = valor índice
 loop:
   CMP A, (len)
                   ;A - Mem[len]
                  ;Termina si index == largo
   JEQ end
                  :B = arr = dirección memoria arr
   MOV B, arr
   ADD B,A ;B = A + B = dirección arr + index
   MOV A, (B) ;A = Mem[B] = Mem[arr+index]
   SHL A, A ; A = SHL A = A * 2
                   ;Mem[arr+index] = Mem[arr+index] * 2
   MOV (B),A
   MOV A, (index)
   ADD A,1
   MOV (index),A
   JMP loop
 end:
    JMP end
```

Otra alternativa para el código anterior. En vez de usar INC B, usamos la suma entre la dirección de memoria arr y el índice index para acceder al elemento de memoria de interés.

```
DATA:

n 5
result 0

CODE:

MOV A,(n) ;A = Mem[n] = valor n

SHR A,A ;A = SHR(A)

JCR is_odd ;if c == 1 goto is_odd

is_even: ;else goto is_even

MOV B,0 ;B = 0

MOV (result),B ;result = 0

JMP end

is_odd:

MOV B,1 ;B = 1

MOV (result),B ;result = 1

end:

JMP end
```

Otro ejemplo consiste en el uso del bit de carry. En este caso, podemos fácilmente detectar si un número es par o impar con un SHR y revisando si hubo carry o no. Notar que no es necesario el uso de **CMP** porque queremos el resultado de la flag C posterior al shift.

Actividad en clase - Ejercicio de Assembly

Elabore, en Assembly, un programa que encuentre la mediana de un arreglo ordenado de números.

Para realizarlo, puede acceder a la <u>siguiente página</u>, donde podrá compilar y ejecutar código Assembly de forma *online*.

Consideremos ahora el siguiente código. Se define una función que no se ejecutará de inmediato, sino que una vez que sea llamada. Esto es lo que llamaremos, dentro de nuestro computador básico, una subrutina. Con ellas, podemos modularizar nuestro código y optimizarlo aún más.

```
def sum(a, b):
    r = 2*a + b
    return r
a = 3
b = 4
c = sum(a, b)
```

Antes de ver su implementación en *hardware*, veamos los elementos que requieren.

Subrutinas - Requisitos

¿Qué elementos necesita una subrutina?

- Parámetros de entrada.
- Valor de retorno.
- Llamada a la subrutina (salto de ida y de vuelta).

Subrutinas - Parámetros de entrada

Al igual que con las variables, necesitamos almacenar los parámetros de una subrutina. Tenemos dos opciones:

- Almacenamiento en los registros: Rápido procesamiento, pero limitado por su cantidad (en nuestra arquitectura solo se podrían definir subrutinas de dos parámetros).
- Almacenamiento en la memoria de datos: Procesamiento más lento por la lectura, pero sin límite de parámetros. Utilizaremos esta implementación.

Subrutinas - Valor de retorno

El valor o valores de retorno de una subrutina presentan la misma logística y opciones que los parámetros de entrada respecto a su almacenamiento:

- Almacenamiento en los registros: Limitado por su cantidad (dos valores de retorno como máximo en este computador).
- Almacenamiento en la memoria de datos: Sin límite para la cantidad de valores de retorno. Utilizaremos esta implementación.


Subrutinas - Llamada y retorno

La llamada y el retorno de una subrutina suponen desafíos distintos.


- La llamada se puede implementar como una instrucción JMP label, siendo label la dirección de memoria de la primera instrucción de la subrutina.
- Para el retorno podemos hacer lo mismo, pero saltando a la instrucción siguiente al llamado. ¿Podemos fijar este valor? ¿Qué pasa si realizamos más de un llamado? Necesitamos almacenar la dirección de la instrucción de retorno.

Necesitamos:

- Conexión entre la memoria de datos y Program Counter para cargar direcciones de instrucciones asociadas a labels.
- Capacidad para almacenar dirección PC+1 en memoria (dirección de retorno).




- Con el Mux DataIn y la señal S_{DIn} escogemos el dato a almacenar en memoria (valor PC+1 o resultado de la ALU).
- Con el Mux PC y la señal S_{PC} ahora escogemos el dato a cargar en el *Program Counter* (literal o la dirección de la instrucción de retorno de la memoria de datos).




Ahora, ¿de dónde proviene la dirección de la instrucción de retorno? ¿Cómo nos aseguramos que su almacenamiento no choque con las variables del programa?

Resolvemos esto a partir de la memoria de stack.




Se hace uso del segmento inferior de la memoria de datos, siendo la **última dirección** la primera de la memoria de *stack*.

Se accede a estas direcciones desde el *Stack Pointer* (SP), contador cuyo valor parte con la última dirección (valor 255 para memorias con bus de 8 bits de direccionamiento).




- La señal Dec_{SP} gatilla el decremento en una unidad el contador SP, lo que aumenta en una unidad el tamaño del stack.
- La señal Inc_{SP} gatilla el incremento en una unidad el contador SP, lo que disminuye en una unidad el tamaño del stack.



Computador básico - Memoria de stack

La figura muestra cómo se divide la memoria de datos con el uso de *stack*. Cabe destacar que el tamaño del *stack* es **dinámico** y su crecimiento dependerá del programa en ejecución.



Por construcción, el *stack pointer* apuntará siempre a **una dirección mayor en una unidad** respecto al tamaño del *stack*. Esto **puede** ser distinto en otras arquitecturas.


Computador básico - Memoria de stack

¿Qué pasa si la memoria de *stack* crece lo suficiente para **chocar** con el segmento donde se definen las variables?

Se genera un stack overflow.



En la práctica, el tamaño de la memoria de stack es fijo y se tiene acceso a la última dirección de memoria disponible, lo que permite detectar el stack overflow antes de que ocurra y matar el programa anticipadamente.



Con las modificaciones en hardware definimos nuevas instrucciones:

- CALL label: Almacena el valor PC+1 en la dirección de memoria SP; reduce en una unidad el contador SP (*i.e.* aumenta el *stack* en una unidad) y carga en PC la dirección de memoria asociada a label (es decir, salta a la primera instrucción de la subrutina).
- Se realiza en un ciclo de clock con la siguiente combinación de señales:

$$S_{Dln} = PC+1; S_{Add} = SP; W = 1; Dec_{SP} = 1; S_{PC} = LIT; L_{PC} = 1$$

Con las modificaciones en *hardware* definimos nuevas instrucciones:

- RET: Aumenta en una unidad el contador SP (*i.e.* disminuye el stack en una unidad); extrae de memoria el valor almacenado en la dirección SP y, finalmente, carga dicho valor en PC (es decir, retorna a la dirección PC+1).
- Se realiza en **dos ciclos de** *clock* de la siguiente forma:

Ciclo 1:
$$lnc_{SP} = 1$$

Ciclo 2:
$$S_{Add} = SP$$
; $S_{PC} = DOUT$; $L_{PC} = 1$

Se requieren dos por la necesidad de actualizar el stack pointer.

Con las modificaciones en *hardware* definimos nuevas instrucciones:

- PUSH Reg: Almacena en la dirección SP el valor del registro Reg. Su composición es similar a CALL, con la salvedad de que en el Mux Dataln se selecciona el resultado de la ALU (que será igual al valor almacenado en Reg).
- POP Reg: Aumenta en una unidad el contador SP y almacena en el registro Reg el dato almacenado en la dirección SP. Su composición es similar a RET, con la salvedad de que el *Program Counter* no actualiza su valor, sino que el registro Reg.

Computador básico - Assembly de subrutinas

Instrucción	Operandos	Operación	Ejemplo
CALL	Dir	Mem[SP] = PC+1; SP; PC = Dir	CALL func
		SP++	
RET	-	PC = Mem[SP]	-
PUSH	А	Mem[SP] = A; SP	-
1 0511	В	Mem[SP] = B; SP	-
		SP++	
DOD	А	A = Mem[SP]	-
POP		SP++	
	В	B = Mem[SP]	-

Ahora que tenemos a nuestra disposición las instrucciones de comparación y saltos, revisemos cómo queda el código de la función sum en Assembly.

En la práctica, PUSH y POP se utilizarán para no perder los valores originales de los registros A y B al momento de ejecutar una subrutina.

```
def sum(a, b):
    r = 2*a + b
    return r
a = 3
b = 4
c = sum(a, b)
```

Programa en pseudocódigo (Python).

```
DATA:
  a 3
  b 4
 c 0
CODE:
 main: ;código principal
    MOV A, (a)
    MOV B, (b)
    CALL sum
    JPM end
  sum: ;subrutina sum
    SHL A,A
    ADD A,B
    RET
  end: ;fin de ejecución
    MOV (c),A
```

Programa en Assembly.

```
DATA:
  fibOne 1
  fibTwo 0
CODE:
  JMP start
  next fibonacci: ;Fibonacci. A = Elemento N y B = Elemento N-1
   MOV (fibTwo),A
                          ;Mem[fibTwo] = Elemento N
   ADD A,B
                         ;A = A + B = Elemento N + 1
   MOV (fibOne),A
                          :Mem[fibOne] = Elemento N + 1
   RET
  start:
   MOV A, (fibOne)
                          ;A = Mem[fibOne] = Elemento N
                          ;B = Mem[fibTwo] = Elemento N - 1
   MOV B, (fibTwo)
   CALL next fibonacci
                          ;Siguiente elemento de la secuencia
   CALL next fibonacci
                          ;Siguiente elemento de la secuencia
   CALL next fibonacci
                          ;Siguiente elemento de la secuencia
  end:
    JMP end
```


Usemos ahora subrutinas para implementar una nueva versión de Fibonacci.


```
DATA:
 fibOne 1
 fibTwo 0
 iterations 3
 currentIteration 0
CODE:
                       ;B = Mem[fibTwo] = Elemento N - 1
 MOV B, (fibTwo)
  JMP loop
 next fibonacci:
   MOV (fibTwo),A
                     ;Mem[fibTwo] = Elemento N
   ADD A,B
                       :A = A + B = Elemento N + 1
   MOV (fibOne),A
                     ;Mem[fibOne] = Elemento N + 1
   MOV B, (fibTwo)
                       ;B = Mem[fibTwo] = Elemento N
    RET
  loop:
   MOV A, (currentIteration)
   CMP A,(iterations)
                        :iterations = currentIteration -> PC = end
    JEQ end
   ADD A,1
   MOV (iterations), A ; iterations += 1
   MOV A, (fibOne) ;A = Mem[fibOne] = Elemento N
    CALL next fibonacci ;Siguiente elemento de la secuencia
    JMP loop
  end:
    JMP end
```

También podemos incorporar ciclos para controlar mejor la cantidad de ejecuciones.

Computador básico - Versión final

Hemos llegado a la versión final del computador básico del curso.





Computador básico - Listado oficial de instrucciones

MOV	Instrucción	Operandos	Opcode	Condición	LPC	L_{A}	L_{B}	SA0,1	$S_B0,1$	$S_{0p}0,1,2$	$S_{Add}0,1$	S_{DIn}	S_{PC}	W	${\tt Inc_{SP}}$	$\mathtt{Dec}_{\mathtt{SP}}$
N. N. 1	MOV											-	-			0
B_iLit											-	-	-			0
A (Dir)			0000010								-	-	-			0
B, (Dar)												=	- 5			0
Chir.) A												3	-			0
CD17 B													-			0
A, (B)													-			0
B, (B)												ALU	-			0
(b) , A 0001010			0001000									-	-			0
ADD A.B A.B A.B ADD B.A OO01000 A.Litt OO01100 A.(Dir) OO01101 OO0100 A.(Dir) OO01100 OO0100 A.(Dir) OO01010 OO01000 OO010000 OO010000 OO010000 OO010000 OO010000 OO010000 OO010000 OO0100000 OO010000 OO010000 OO010000 OO010000 OO010000 OO010000 OO0100000 OO0100000 OO0100000 OO0100000 OO0100000 OO0100000 OO01000000 OO010000000 OO01000000 OO010000000 OO01000000 OO010000000 OO010000000 OO0100000000												-	-			0
B.A											В	ALU	-			0
A Lit	ADD	A,B						A			100	-	-			0
A (O)											-		5			0
A, (B) 0001111 0 1 0 A DOUT ADD B 0 0 0 0 0 0 0 0												-	-			0
Oil Oil		A,(Dir)	0001110									-	-			0
SUB		A,(B)	0001111					A	DOUT	ADD	В	-	-			0
B_A			0010000					A			LIT	ALU	-			0
A_List	SUB	A,B	0010001			1	0	A	В	SUB	-	-	-			0
A_(O)		B,A	0010010		0	0	1	A	В	SUB	-	-	-	0	0	0
A, (B)		A,Lit	0010011		0	1	0	A	LIT	SUB		-	-	0	0	0
Object O		A,(Dir)	0010100		0	1	0	A	DOUT	SUB	LIT	-	-	0	0	0
AND		A,(B)	0010101		0	1	0	A	DOUT	SUB	В	-	4	0	0	0
B_A		(Dir)	0010110		0	0	0	A	В	SUB	LIT	ALU	-	1	0	0
A_Lit	AND	A,B	0010111		0	1	0	A	В	AND	-	-	-	0	0	0
A (O17)		B,A	0011000		0	0	1	A	В	AND	-	-	-	0	0	0
A, (B)		A,Lit	0011001		0	1	0	A	LIT	AND	-	-	-	0	0	0
Object O		A, (Dir)	0011010		0	1	0	A	DOUT	AND	LIT	-	-	0	0	0
OR		A, (B)	0011011		0	1	0	A	DOUT	AND	В	-	-	0	0	0
OR A,B 0011101 0 1 0 A B OR 0 0 0 B,A 0011101 0 0 1 0 A B OR 0 0 0 A,Lit 0011111 0 0 1 0 A LIT OR 0 0 0 A,CD 011111 0 0 1 0 A LIT OR 0 0 0 A,CD 011111 0 0 1 0 A LIT OR 0 0 0 A,CD 011111 0 0 1 0 A DOUT OR LIT 0 0 0 A,CD 0100001 0 1 0 A DOUT OR LIT 0 0 0 A,CD 0100010 0 0 1 0 A B OR LIT ALU - 1 0 0 B,A 0100101 0 0 1 0 A B OR LIT ALU - 1 0 0 B,A 0100101 0 0 0 1 A B NOT LIT ALU - 1 0 0 CD 01 0 0 A B NOT LIT ALU - 1 0 0 A,B 0100101 0 1 0 A B NOT LIT ALU - 1 0 0 B,A 0100101 0 1 0 A B NOT LIT ALU - 1 0 0 A,B 010011 0 1 0 A B NOT LIT ALU - 1 0 0 B,A 0100101 0 1 0 A B NOT LIT ALU - 1 0 0 A,LIT 010000 0 0 1 A B NOT LIT ALU - 1 0 0 A,LIT 010000 0 0 1 A B NOT LIT ALU - 0 0 0 A,CD 010011 0 0 1 0 A B NOT LIT ALU - 0 0 0 A,CD 010011 0 0 1 0 A B NOT LIT NOR 0 0 0 A,CD 0100101 0 1 0 A DOUT NOR LIT - 0 0 0 A,CD 0101010 0 1 0 A DOUT NOR LIT - 0 0 0 A,CD 010101 0 0 1 0 A DOUT NOR LIT - 0 0 0 CD 1 0 0 A B NOR NOR 0 0 0 CD 1 0 0 A B NOR LIT ALU - 1 0 SHL A,A 010110 0 0 1 0 A B NOR NOR LIT ALU - 1 0 SHL A,A 010110 0 0 0 1 A B NOR NOR LIT ALU - 1 0 SHL A,A 010110 0 0 0 0 A B SHR LIT ALU - 1 0 SHR A,A 010110 0 0 0 0 A B SHR LIT ALU - 1 0 CD 0 0 0 0 0 0 A B SHR LIT ALU - 1 0 CD 0 0 0 0 0 0 A B SHR LIT ALU - 1 0 CD 0 0 0 0 0 0 0 A B SHR LIT ALU - 1 0 CD 0 0 0 0 0 0 A B SHR LIT ALU - 1 0 CD 0 0 0 0 0 0 A B SHR LIT ALU - 1 0 CD 0 0 0 0 0 0 0 A B SHR LIT ALU - 1 0 0 CD 0 0 0 0 0 0 A B SHR LIT ALU - 1 0 0 CD 0 0 0 0 0 0 A B SHR LIT ALU - 1 0 0 CD 0 0 0 0 0 0 A B SHR LIT ALU - 1 0 0 CD 0 0 0 0 0 A B SHR LIT ALU - 1 0 0 CD 0 0 0 0 0 0 A B SHR LIT ALU - 1 0 0 CD 0 0 0 0 0 0 A B SHR LIT ALU - 1 0 0 CD 0 0 0 0 0 0 A B SHR LIT ALU - 1 0 0 CD 0 0 0 0 0 0 A B SHR LIT ALU - 1 0 0 CD 0 0 0 0 0 0 A B SHR LIT ALU - 1 0 0 0 CD 0 0 0 0 0 0 A B SHR LIT ALU - 1 0 0 0 0 0 CD 0 0 0 0 0 A B SHR LIT ALU - 1 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0			0011100		0	0	0	A	В	AND	LIT	ALU	-	1	0	0
B_A	OR				0	1	0	A	В	OR	-	2	- 2	0	0	0
A					0	0	1				-	-	-	0	0	0
A, (Dir)					0	1	0				-	-	-		0	0
A, (B)					0	1	0	A		OR.	LIT	-	-		0	0
						1	0					-			0	0
NOT A,A 0100011 0 1 0 A - NOT - - 0 0 0 0 0 0 0 0						0						ALU	-		0	0
B,A	NOT															0
Colif)																0
XOR									В		LIT	ALU	- 5			0
B,A	YOR															0
A_Lit 0101000	AUI											- 8	- 3			0
$ \begin{array}{c ccccccccccccccccccccccccccccccccccc$																0
A, (B)												-				0
$ \begin{array}{c ccccccccccccccccccccccccccccccccccc$												-	- 0			0
SHL												AT II	-			0
B,A 0101101 0 0 1 A - SHL - - - 0 0 0	our												-			0
(Dir) 0101110 0 0 0 A B SHL LIT ALU - 1 0 SHR A, A 0101111 0 1 0 A - SHR - - - 0 0 0 B, A 0110000 0 0 0 1 A - SHR - - - 0 0 0 (Dir) 0110001 0 0 0 A B SHR LIT ALU - 1 0 INC B 0110010 0 0 A B SHR LIT ALU - 1 0 0 0 A B SHR LIT ALU - 1 0 0 A B SHR LIT ALU - 1 0 0 A B SHR LIT ALU - 1 0 0 <	OILL											- 0	- 1			0
SHR A,A 0101111												AT II	- 5			0
B,A 0110000 0 0 1 A - SHR 0 0 (Dir) 0110001 0 0 0 A B SHR LIT ALU - 1 0 INC B 0110010 0 0 1 NE B ADD 0 0	GUD												3			0
$ \begin{array}{c ccccccccccccccccccccccccccccccccccc$	DIII												-			0
INC B 0110010 0 0 1 ONE B ADD 0 0													-			0
	TNG												-			0
	INC												-			0
(Dir) 0110100 0 0 1 ONE DOUT ADD LIT ALU - 1 0													-			0

NOP		1001001		0	0	0			-				0	0	0
		1001000		0	0	1	ZERO	DOUT	ADD	SP	-	-	0	0	0
POP	В	1000011		0	0	0	-	15	1.5	1.5	-	-	0	1	0
		1000111		0	1	0	ZERO	DOUT	ADD	SP	-	-	0	0	0
POP	A	1000011		0	0	0	-	-	-	-	-	-	0	1	0
	В	1000110		0	0	0	ZERO	В	ADD	SP	ALU	-	1	0	1
PUSH	A	1000101		0	0	0	A	ZERO	ADD	SP	ALU	-	1	0	1
		1000100		1	0	0	-		-	SP	-	DOUT	0	0	0
RET		1000011		0	0	0	-	-	-		-	-	0	1	0
CALL	Dir	1000010		1	0	0	-	-	-	SP	PC	LIT	1	0	1
JOV	Dir	1000001	V=1	1	0	0	-	-	-	-	-	LIT	0	0	0
JCR	Dir	1000000	C=1	1	0	0	-	-		-	-	LIT	0	0	0
JLE	Dir	0111111	N=1 o Z=1	1	0	0	-	-			-	LIT	0	0	0
JGE	Dir	0111110	N=O	1	0	0	-	-		-	-	LIT	0	0	0
JLT	Dir	0111101	N=1	1	0	0				141		LIT	0	0	0
JGT	Dir	0111100	N=0 y Z=0	1	0	0	-		-	14		LIT	0	0	0
JNE	Dir	0111011	Z=0	î	0	0				10	- 6	LIT	0	0	0
JEQ	Dir	0111010	Z=1	1	0	0	-					LIT	0	0	0
JMP	Dir	0111001		1	0	0	-	-	-	-		LIT	0	0	0
	A, (B)	0111000		0	0	0	A	DOUT	SUB	В	-	-	0	0	0
	A,(Dir)	0110111		0	0	0	A	DOUT	SUB	LIT			0	0	0
CMP	A,B A,Lit	0110101 0110110		0	0	0	A A	B LIT	SUB	-	-	-	0	0	0

```
DATA:
  temp 0
CODE:
  MOV A.3
  MOV B,5
  CALL swap_3
  JMP end
                 :Intercambio con uso de memoria.
  swap 1:
   MOV (temp), A ; Mem[temp] = A
   MOV A,B
                 A = B
   MOV B, (temp) ;B = Mem[temp] = A
    RET
  swap 2:
                 :Intercambio con uso de stack
    PUSH A
                 :Mem[SP] = A
    MOV A,B
                 :A = B
                 ;B = Mem[SP] = A (2 ciclos)
   POP B
    RET
  swap 3:
                 ;Intercambio con XOR
   XOR A,B
                A = A XOR B
                 ;B = (A XOR B) XOR B = A XOR (B XOR B) = A XOR 0 = A
   XOR B,A
                 ; A = (A XOR B) XOR A = B XOR (A XOR A) = B XOR 0 = B
   XOR A,B
    RET
                 ;Intercambio con ADD y SUB
  swap 4:
   ADD A,B
                A = A + B
               ;B = (A + B) - B = A
   SUB B,A
   SUB A,B
                 A = (A + B) - A = B
   RET
  end:
    JMP end
```

Un ejemplo útil es el intercambio de registros. Este se puede realizar de varias maneras, como se ilustra a continuación. Basta con que cambiemos el *label* de la instrucción **CALL** para cambiar el método, pero todos hacen lo mismo.

```
DATA:
  a 7
  b 9
  r 0
CODE:
  loop:
    MOV A,(a)
    CMP A,0
    JEQ end
                         A == 0 \rightarrow PC = end
    AND A,1
                   ;A \% 2 == 0 -> PC = avoid addition
    CMP A,0
    JEO avoid addition ; No se suma si el número es par
    MOV A,(r)
    MOV B, (b)
    ADD (r)
                         :r += b
    avoid addition:
      MOV A,(a)
      SHR (a)
                         ;a = a // 2 (división entera por 2)
      MOV A, (b)
                         b = b * 2
      SHL (b)
      JMP loop
    end:
      JMP end
```

Otro ejemplo interesante es la multiplicación rusa. Esta se basa en realizarla solo a partir de shifts left/right sobre los operandos (i.e. dividir y multiplicar por dos). Pueden ver en los siguientes enlaces cómo funciona y por qué funciona.

```
CODE:
 MOV A, 3
 MOV B,7
 PUSH A
             |Mem[255]| = A = 3
             |Mem[254]| = B = 7
 PUSH B
 MOV A,0
 NOT A,A
           A = 111111111
 SUB A,1 ;A = 111111110
 MOV B, A ; B = 111111110
 MOV A, (B) ; A = Mem[111111110] = Mem[254] = 7
 INC B
        ;B = 111111111
 MOV B, (B) ; B = Mem[11111111] = Mem[255] = 3
```

Podemos acceder directamente a la memoria de stack a través de direccionamiento indirecto. Esto es útil al momento de querer corroborar que nuestras instrucciones de stack y subrutinas escriben en memoria correctamente.

```
DATA:
  sum 0
CODE:
 MOV A,3
 MOV B,7
  PUSH A
             |Mem[255]| = A = 3
         |Mem[254]| = B = 7
  PUSH B
  CALL func ;Mem[253] = Ret. Dir = PC+1 = 5
             ;No se ejecuta por cambio de retorno.
  POP A
  POP B ; No se ejecuta por cambio de retorno.
  JMP end
             ;Se ejecuta después de RET.
  func:
   ADD (sum)
   POP A ; A = Mem[253] = 5
             ;PC = Mem[254] = 7
   RET
  end:
```

Por último, podemos modificar la dirección de retorno de una subrutina para cambiar el retorno. En el ejemplo, el POP de func hace que se pierda el retorno y PC carga el valor ingresado de B que, en este caso, hace que se ejecute JMP end y finalice el programa antes.

Ahora, veremos algunos ejercicios.

Estos se basan en preguntas de tareas y pruebas de semestres anteriores, por lo que nos servirán de preparación para las evaluaciones.



Si se elimina la instrucción CMP del computador básico, ¿cómo deben modificarse las instrucciones de salto, sin alterar el *hardware*, para que estas no dependan del resultado de la última instrucción ejecutada? Escriba detalladamente todas las modificaciones necesarias y sus implicancias. Asuma que solo es necesario resolver el caso de la comparación de los registros *A* y *B* y que no es posible sobreescribir los registros para realizar la comparación.

Interrogación 1, 2017-1

Modifique la arquitectura del computador básico para que el registro Status se actualice solo después de la ejecución de una instrucción CMP.

Examen, 2017-1

Modifique el *hardware* del computador básico para que las instrucciones RET y POP tomen un solo ciclo.

Interrogación 1, 2018-1

Modifique la arquitectura del computador básico para implementar las instrucciones NOT B,B; SHL B,B y SHR B,B; es decir, que realice las operaciones de un operando sobre el registro B y almacene el resultado en él.

Examen, 2023-1

Modifique la arquitectura del computador básico para implementar las instrucciones MOV A, (B+offset) y MOV (B+offset), A, i.e. instrucciones de direccionamiento indirecto con registro B y offset, siendo este último un literal que puede ser positivo o negativo.

Examen, 2023-1

Asuma que se agregan las instrucciones anteriores a la ISA del computador básico y se ejecuta el programa adjunto. Señale solo los valores finales de los registros *A*, *B* y SP. Asuma que todos los registros son de 8 bits y que SP se interpreta como un número positivo.

```
MOV A,2
                  ; Dirección Mem. Instr.: 0x00
MOV B,5
                  ; Dirección Mem. Instr.: 0x01
PUSH A
                  ; Dirección Mem. Instr.: 0x02
PUSH B
                  : Dirección Mem. Instr.: 0x03
                  ; Dirección Mem. Instr.: 0x04
CALL mystery
POP B
                  : Dirección Mem. Instr.: 0x05-0x06
POP A
                  ; Dirección Mem. Instr.: 0x07-0x08
JMP end
                  ; Dirección Mem. Instr.: 0x09
mystery:
  MOV B,0
                  : Dirección Mem. Instr.: 0x0A
  NOT B,B
                  : Dirección Mem. Instr.: 0x0B
  MOV A.(B + -2) : Dirección Mem. Instr.: 0x0C
  ADD A,4
                  ; Dirección Mem. Instr.: 0x0D
  MOV (B + -2), A ; Dirección Mem. Instr.: 0x0E
  RET
                  ; Dirección Mem. Instr.: 0x0F-0x10
end:
```

Antes de terminar

¿Dudas?

¿Consultas?

¿Inquietudes?

¿Comentarios?





Arquitectura de Computadores

Clase 5 - Saltos y Subrutinas

Profesor: Germán Leandro Contreras Sagredo

Anexo - Resolución de ejercicios

¡Importante!

Estos ejercicios pueden tener más de un desarrollo correcto. Las respuestas a continuación no son más que soluciones que **no excluyen** otras alternativas igual de correctas.



Actividad en clase - Respuesta

Elabore, en Assembly, un programa que encuentre la mediana de un arreglo ordenado de números.

- Si existe una cantidad impar de elementos, obtenemos el índice de la mitad del arreglo.
- Si existe una cantidad par de elementos, sumamos los números centrales y realizamos la división entera por 2.

```
DATA:
  len 6
  median 0
  arr -5
       11
       20
CODE:
  MOV A, (len)
  AND A,1
                       ; Si A es par, AND A,1 == 0. Si es impar, AND A,1 == 1
  CMP A.0
  JEO even arr
  odd arr:
    MOV A, (len)
    SHR A,A
                       ; Índice mitad del arreglo = len // 2
                       : B = dirección arr
    MOV B, arr
    ADD B,A
                       : B = dirección arr + len // 2
                       : A = mediana del arreglo
    MOV A, (B)
    JMP end
  even_arr:
    MOV A, (len)
    SHR A,A
                       : Primer elemento de la mediana = len // 2 - 1
    SUB A,1
    MOV B, arr
                       ; B = dirección arr
    ADD B,A
                       ; B = dirección arr + len // 2 - 1
    MOV A, (B)
                       : A = primer elemento de la mediana
    INC B
                       ; Segundo elemento de la mediana = len // 2
                       : A = (primer elemento + segundo elemento) // 2
    ADD A, (B)
    SHR A,A
  end:
    MOV (median),A
                        ; Escribimos en la variable "median" el resultado
```

Si se elimina la instrucción CMP del computador básico, ¿cómo deben modificarse las instrucciones de salto, sin alterar el *hardware*, para que estas no dependan del resultado de la última instrucción ejecutada? Escriba detalladamente todas las modificaciones necesarias y sus implicancias. Asuma que solo es necesario resolver el caso de la comparación de los registros *A* y *B* y que no es posible sobreescribir los registros para realizar la comparación.

Respuesta en la siguiente diapositiva.

Como no queremos que los saltos dependan de la instrucción anterior (esto es, que no se basen en cómo haya quedado el registro STATUS después de ejecutarlas), lo que se necesita hacer es añadir más operaciones a las instrucciones de salto. En este caso, lo que se debe hacer es replicar la instrucción CMP dentro de la ejecución que realizan los saltos. De esta forma, cada instrucción de salto contendrá dos *opcodes*:

- 1) El primero corresponderá al mismo que solía tener CMP: Se realiza la resta entre los registros A y B sin guardar el resultado para poder actualizar las flags del registro STATUS.
- 2) El segundo será el que tenía cada salto originalmente.


De esta forma, el cambio sustancial que se genera es que las instrucciones de salto toman dos ciclos en vez de uno.

Modifique la arquitectura del computador básico para que el registro Status se actualice solo después de la ejecución de una instrucción CMP.

Basta con crear una nueva señal $L_{\rm Stat}$ (señal de carga del registro Status) que se active si, y solo si el opcode de la instrucción corresponde a CMP. En otro caso, la unidad de control debe encargarse de transmitir $L_{\rm Stat}=0$.

Modifique el *hardware* del computador básico para que las instrucciones RET y POP tomen un solo ciclo.

Basta con conectar SP a un sumador que recibe, además, la señal $Inc_{\rm SP}$. Solo cuando activamos esta señal es cuando necesitamos acceder a la siguiente dirección de la memoria stack, por lo que la dirección se modifica directamente y no requerimos esperar un ciclo para el ajuste del puntero.




Modifique la arquitectura del computador básico para implementar las instrucciones NOT B,B; SHL B,B y SHR B,B; es decir, que realice las operaciones de un operando sobre el registro B y almacene el resultado en él.

Respuesta en la siguiente diapositiva.

Solución: Para esta implementación, se conecta el registro B con el componente Mux A para utilizar el valor de B como *input* de las operaciones NOT, SHL y SHR de la ALU:

De esta forma, sin modificar ni agregar las señales de la microarquitectura, se tiene la siguiente tabla para la ejecución de las instrucciones implementadas:

Instrucción	LA	LB	L _{PC}	W	$\mathtt{Inc}_\mathtt{SP}$	Decsp	SA	SB	Sop	SAdd	S_{DIn}	S_{PC}
NOT B,B	0	1	0	0	0	0	В	-	NOT	-	-	-
SHL B,B	0	1	0	0	0	0	В	-	SHL	-	-	-
SHR B,B	0	1	0	0	0	0	В	-	SHR	н	-	-




Modifique la arquitectura del computador básico para implementar las instrucciones MOV A, (B+offset) y MOV (B+offset), A, i.e. instrucciones de direccionamiento indirecto con registro B y offset, siendo este último un literal que puede ser positivo o negativo.

Respuesta en la siguiente diapositiva.

Solución: Para la implementación de estas instrucciones, se puede modificar la conexión entre el registro B y el componente Mux Address para que ahora reciba el resultado de un sumador que sume el valor de B y el literal de la instrucción:

De esta forma, no es necesario agregar ni modificar señales, pero sí asegurar que el compilador haga uso del literal 0 para las instrucciones MOV A,(B) y MOV (B),A. La tabla de señales es como sigue para las instrucciones implementadas:

Instrucción	LA	LB	L_{PC}	W	Inc_{SP}	Decsp	SA	SB	SOP	S_{Add}	SDIn	SPC
MOV A, (B+offset)	1	0	0	0	0	0	0	DOUT	ADD	B+offset	-	-
MOV (B+offset),A	0	0	0	0	0	0	A	0	ADD	B+offset	ALU	-



Asuma que se agregan las instrucciones anteriores a la ISA del computador básico y se ejecuta el programa adjunto. Señale solo los valores finales de los registros *A*, *B* y SP. Asuma que todos los registros son de 8 bits y que SP se interpreta como un número positivo.

```
MOV A,2
                  ; Dirección Mem. Instr.: 0x00
MOV B,5
                  ; Dirección Mem. Instr.: 0x01
PUSH A
                  ; Dirección Mem. Instr.: 0x02
PUSH B
                  : Dirección Mem. Instr.: 0x03
CALL mystery
                  : Dirección Mem. Instr.: 0x04
POP B
                  : Dirección Mem. Instr.: 0x05-0x06
POP A
                  ; Dirección Mem. Instr.: 0x07-0x08
JMP end
                  ; Dirección Mem. Instr.: 0x09
mystery:
 MOV B,0
                  : Dirección Mem. Instr.: 0x0A
  NOT B,B
                  : Dirección Mem. Instr.: 0x0B
  MOV A.(B + -2) : Dirección Mem. Instr.: 0x0C
  ADD A,4
                  ; Dirección Mem. Instr.: 0x0D
  MOV (B + -2), A ; Dirección Mem. Instr.: 0x0E
  RET
                  ; Dirección Mem. Instr.: 0x0F-0x10
end:
```

Respuesta en la siguiente diapositiva.

Solución: Se señalan los puntos clave en donde se modifican los registros y por qué:

- 1. Antes de CALL mystery, se tiene A = 2, B = 5 por las instrucciones MOV y SP = 253 por las instrucciones PUSH.
- 2. Posterior a CALL mystery, se tiene A = 2, B = 5 y SP = 252 por el almacenamiento de la dirección de retorno 0x05 (PC+1).
- 3. Antes de RET, se tiene A = 9, B = 255 y SP = 252. En este punto, lo que se hizo fue modificar la dirección de retorno de la subrutina mystery, apuntando ahora a la instrucción JMP end.
- 4. Posterior a RET, el programa termina sin ejecutar las instrucciones POP y se tiene A = 9, B = 255 y SP = 253.